

Variable Range PWM A/D for COP820CJ

National Semiconductor
Application Note 815
Kevin Daugherty
March 1993



Variable Range PWM A/D for COP820CJ

This application note is an extension of AN-607. The basic difference between the previous version and this new approach (patent pending) is that the microcontroller can vary the computing range as required by the signal being measured. This can be done totally in software without any additional hardware.

Microcontrollers such as the COP820CJ that have an on board comparator allow for a very cost effective A/D (refer to *Figure 1*). Note that there are two back to back diodes in the circuit diagram in addition to the input resistor and R/C network. The diodes provide a speed up path that assists in initializing the capacitor to equal the unknown input voltage prior to a conversion.

Since the on board comparator has an input limitation of 0.4V to $V_{CC} - 1.5V$, time and resolution would be wasted if the R/C network was driven with a PWM signal that was either at GND or V_{CC} the entire sample time. The waveforms (*Figure 2*) used in this example, have a duty cycle for the high and low pulses. High pulses consist of 8 instruction cycles at V_{CC} and 16 cycles low. Thus providing an upper range of 16/24 (V_{CC}) or 3.30V for full counts with $V_{CC} = 5V$. Low pulses consist of 5 cycles at V_{CC} and 19 cycles at ground. Thus providing a lower range 5/24 (V_{CC})

or 1.041V for zero counts. Equation 1 below illustrates various measurements:

Let:

$$V_L = V_{CC} \text{ (low pulse duty cycle) ; average voltage}$$

$$V_H = V_{CC} \text{ (high pulse duty cycle) ; average voltage}$$

EQU. 1:

$$V_{IN} = V_L + (V_H - V_L) \times (\text{Ton counts/Total counts})$$

The flow chart (*Figure 3*) and code listing (*Figure 4*) uses 100 counts over an input range of 1.0V to 3.30V for 23 mV per count resolution. Many variations of this technique are possible to meet almost any range or speed desired. Simply change the number of counts or the high and low duty cycles. For example, with a known input range of 1.25V–2.5V the low pulse would consist of 6 cycles high and 18 cycles low. High pulses would consist of 12 cycles high and 12 cycles low. Another alternative could involve a rough measurement to determine the approximate range of the input. Following this, duty cycles with an average voltage just below (low pulse) and just above (high pulse) the initial measurement can be used to give a high resolution result relatively fast.




FIGURE 1. Basic Circuit

TL/DD/11419-1




FIGURE 2. Pulse Waveforms

TL/DD/11419-2




FIGURE 3. PWM Flow Diagram

TL/DD/11419-3

```

;FILE VARPWM.TXT, COP822CJ, KEVIN DAUGHERTY 3-30-92
;THIS ROUTINE PROVIDES FOR LIMITED RANGE ON THE INPUT TO AN
;ONBOARD COMPARATOR .
;THE INPUT RANGE FOR THE A/D IS 1V
;TO 3.25V. THIS REQUIRES A HIGH PULSE DUTY CYCLE = 8 LOW
;AND 16 HIGH CYCLES, AND A LOW PULSE DUTY CYCLE = 5 HIGH
;AND 19 LOW CYCLES. L1=(-) COMPARATOR INPUT CONNECTED TO
;R/C NETWORK, L2=(+) INPUT FOR Vin, L3 DRIVES R/C.
;100 COUNTS OF RESOLUTION ARE STORED IN RAM LOCATION 00.
;ZERO COUNTS EQUALS 1V AND 100 COUNTS EQUALS 3.25V
;Vcc=4.75V. WITH THE RANGE PROPORTIONAL TO THE SUPPLY
;VOLTAGE.
.CHIP 820
CNTRL2=0CC
LDATA=0D0
CMPSL=0B7
LCONF=0D1
TON=0F2
TOTAL=0F0
;
;          SBIT 4,CNTRL2 ;SET COMPARATOR ENABLE BIT
LD LCONF,#00 ;SETUP L1&2 AS INPUTS
LD LDATA,#00 ;TRISTATE L INPUTS
CONV:      LD A,#02      ;USE FOR COUNTING TOTAL LOOPS
LD 0F1,#02 ;TOTAL LOOP COUNTER
LD TOTAL,#064 ;PRELOAD TOTAL =100 COUNTS
LD TON,#064 ;PRELOAD TON =100 COUNTS
LD 0FE,#0D0 ;INIT. B REG TO POINT TO Ldata REG
SBIT 3,LDATA ; L3=HIGH
SBIT 3,LCONF ; L3=OUTPUT
LOOP:      IFBIT 3,CNTRL2 ;TEST COMPARATOR OUTPUT BIT
JP HIGH   ;JUMP IF COMPARATOR= HIGH
NOP
NOP        ;EQUALIZE TIME FOR SET AND RESET
SBIT 3,[B] ;DRIVE L3 HIGH 5 PULSES
DRSZ TON   ;DECREMENT Ton WHEN APPLYING NEG. REF.
NOP
RBIT 3,[B] ;DRIVE L3 LOW WHEN COMPARATOR IS LOW.
NOP        ;EQUALIZE HIGH AND LOW LOOP CYCLES
NOP
JP COUNT   ;JUMP TO COUNT UNLESS TON REACHES ZERO
HIGH:      RBIT 3,[B] ;RESET L3 FOR TOTAL OF 8 CYCLES
NOP
NOP
NOP
NOP
NOP
NOP
NOP
SBIT 3,[B] ;DRIVE L3 HIGH FOR TOTAL OF 16 CYCLES
NOP
NOP
COUNT:    DRSZ TOTAL ;DEC. TOTAL COUNTS EACH LOOP
JP LOOP    ;JUMP UNLESS TOTAL CNTS.=0
RBIT 3,LCONF ;TRISTATE L3 TO MINIMIZE ERROR
RBIT 3,[B]  ; "
IFEQ A,0F1 ;CHECK FOR 2nd CONVERSION COMPLETE
JP RELOAD  ;IF TRUE
JP DEC     ;OTHERWISE JUMP TO DEC
RELOAD:    LD TON,#064 ;RELOAD TON FOR START OF NEXT CONV.
LD TOTAL,#064 ;SYNC TON AND TOTAL COUNTS
DEC:       SBIT 3,[B] ;SET L3 HIGH
SBIT 3,LCONF ;RESTORE L3 AS OUTPUT
DRSZ 0F1    ;DECREMENT TOTAL LOOP UNTIL ZERO.
JMP LOOP    ;DONE WHEN 0F1 IS ZERO.
LD A,TON    ;LOAD A WITH Ton RESULT
X A,00      ;STORE RESULTS IN RAM 00
.END

```

TL/DD/11419-4

FIGURE 4. Code Listing

LIFE SUPPORT POLICY

NATIONAL'S PRODUCTS ARE NOT AUTHORIZED FOR USE AS CRITICAL COMPONENTS IN LIFE SUPPORT DEVICES OR SYSTEMS WITHOUT THE EXPRESS WRITTEN APPROVAL OF THE PRESIDENT OF NATIONAL SEMICONDUCTOR CORPORATION. As used herein:

1. Life support devices or systems are devices or systems which, (a) are intended for surgical implant into the body, or (b) support or sustain life, and whose failure to perform, when properly used in accordance with instructions for use provided in the labeling, can be reasonably expected to result in a significant injury to the user.
2. A critical component is any component of a life support device or system whose failure to perform can be reasonably expected to cause the failure of the life support device or system, or to affect its safety or effectiveness.



National Semiconductor Corporation
2900 Semiconductor Drive
P.O. Box 58090
Santa Clara, CA 95052-8090
Tel: 1(800) 272-9959
TWX: (910) 339-9240

National Semiconductor GmbH
Livny-Gargan-Str. 10
D-82256 Fürstenfeldbruck
Germany
Tel: (81-41) 35-0
Telex: 527849
Fax: (81-41) 35-1

National Semiconductor Japan Ltd.
Sumitomo Chemical
Engineering Center
Bldg. 7F
1-7-1, Nakase, Mihama-Ku
Chiba-City,
Chiba Prefecture 261
Tel: (043) 299-2300
Fax: (043) 299-2500

National Semiconductor Hong Kong Ltd.
13th Floor, Straight Block,
Ocean Centre, 5 Canton Rd.
Tsimshatsui, Kowloon
Hong Kong
Tel: (852) 2737-1600
Fax: (852) 2736-9960

National Semicondutores Do Brazil Ltda.
Rue Deputado Lacorda Franco
120-3A
Sao Paulo-SP
Brazil 05418-000
Tel: (55-11) 212-5066
Telex: 391-1131931 NSBR BR
Fax: (55-11) 212-1181

National Semiconductor (Australia) Pty, Ltd.
Building 16
Business Park Drive
Monash Business Park
Nottingham, Melbourne
Victoria 3168 Australia
Tel: (3) 558-9999
Fax: (3) 558-9998

National does not assume any responsibility for use of any circuitry described, no circuit patent licenses are implied and National reserves the right at any time without notice to change said circuitry and specifications.